# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=plecost
pkgver=104.4895e34
pkgrel=6
groups=('blackarch' 'blackarch-webapp' 'blackarch-fingerprint')
pkgdesc='Wordpress finger printer Tool.'
arch=('any')
url='https://github.com/iniqua/plecost'
license=('BSD')
depends=('python' 'python-lxml' 'python-termcolor' 'python-aiohttp'
         'python-simplejson' 'python-chardet' 'python-beautifulsoup4'
         'python-sphinx_rtd_theme')
makedepends=('git' 'python-setuptools')
source=("git+https://github.com/iniqua/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

package() {
  cd $pkgname

  python setup.py install --root="$pkgdir" --prefix=/usr -O1

  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.md CHANGELOG.md
}

